grammar FieldSpecifierNotation;

@header {
/*
 * Licensed to the Apache Software Foundation (ASF) under one
 * or more contributor license agreements.  See the NOTICE file
 * distributed with this work for additional information
 * regarding copyright ownership.  The ASF licenses this file
 * to you under the Apache License, Version 2.0 (the
 * "License"); you may not use this file except in compliance
 * with the License.  You may obtain a copy of the License at
 *
 *     http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package org.apache.beam.sdk.schemas.parser.generated;
}

fieldSpecifier: dotExpression;

dotExpression: dotExpressionComponent ('.' dotExpressionComponent)*;

dotExpressionComponent: qualifiedComponent # qualifyComponent
                      | IDENTIFIER  # simpleIdentifier
                      | WILDCARD     # wildcard
                      ;

qualifiedComponent: IDENTIFIER qualifierList
                  ;

qualifierList: arrayQualifier (qualifierList)* # arrayQualifierList
             | mapQualifier (qualifierList)*   # mapQualifierList
             ;

arrayQualifier: '[]'
              | '[*]'
              ;

mapQualifier: '{}'
            | '{*}'
            ;

IDENTIFIER: [a-zA-Z0-9_]+ ;

WILDCARD: '*';

WS: [ \t\n\r]+ -> skip ;